NAG Library Routine Document

G05SBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05SBF generates a vector of pseudorandom numbers taken from a beta distribution with parameters a and b.

2 Specification

```
SUBROUTINE G05SBF(N, A, B, STATE, X, IFAIL)
INTEGER           N, STATE(lstate), IFAIL
double precision  A, B, X(N)
```

3 Description

The beta distribution has PDF (probability density function)

$$f(x) = \frac{\Gamma(a+b)}{\Gamma(a)\Gamma(b)} x^{a-1}(1-x)^{b-1} \quad \text{if } 0 \leq x \leq 1;\ a, b > 0,$$

$$f(x) = 0 \quad \text{otherwise.}$$

One of four algorithms is used to generate the variates depending on the values of a and b. Let α be the maximum and β be the minimum of a and b. Then the algorithms are as follows:

(i) if $\alpha < 0.5$, Johnk's algorithm is used, see for example Dagpunar (1988). This generates the beta variate as $u_1^{1/a} / \left(u_1^{1/a} + u_2^{1/b} \right)$, where u_1 and u_2 are uniformly distributed random variates;

(ii) if $\beta > 1$, the algorithm BB given by Cheng (1978) is used. This involves the generation of an observation from a beta distribution of the second kind by the envelope rejection method using a log-logistic target distribution and then transforming it to a beta variate;

(iii) if $\alpha > 1$ and $\beta < 1$, the switching algorithm given by Atkinson (1979) is used. The two target distributions used are $f_1(x) = \beta x^{\beta}$ and $f_2(x) = \alpha(1-x)^{\beta-1}$, along with the approximation to the switching parameter of $t = (1-\beta)/(\alpha+1-\beta)$;

(iv) in all other cases, Cheng's BC algorithm (see Cheng (1978)) is used with modifications suggested by Dagpunar (1988). This algorithm is similar to BB, used when $\beta > 1$, but is tuned for small values of a and b.

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05SBF.

4 References

Atkinson A C (1979) A family of switching algorithms for the computer generation of beta random variates *Biometrika* **66** 141–5

Cheng R C H (1978) Generating beta variates with nonintegral shape parameters *Comm. ACM* **21** 317–322

Dagpunar J (1988) *Principles of Random Variate Generation* Oxford University Press

Hastings N A J and Peacock J B (1975) *Statistical Distributions* Butterworth

5 Parameters

1: N – INTEGER *Input*

On entry: n, the number of pseudorandom numbers to be generated.

Constraint: N ≥ 0.

2: A – ***double precision*** *Input*

On entry: a, the parameter of the beta distribution.

Constraint: A > 0.0.

3: B – ***double precision*** *Input*

On entry: b, the parameter of the beta distribution.

Constraint: B > 0.0.

4: STATE($lstate$) – INTEGER array *Communication Array*

Note: $lstate$ = LSTATE, the dimension of STATE as supplied to the initialization routine G05KFF or G05KGF.

On entry: contains information on the selected base generator and its current state.

On exit: contains updated information on the state of the generator.

5: X(N) – ***double precision*** array *Output*

On exit: the n pseudorandom numbers from the specified beta distribution.

6: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, N < 0.

IFAIL $= 2$

On entry, A ≤ 0.0.

IFAIL $= 3$

On entry, B ≤ 0.0.

IFAIL $= 4$

On entry, STATE vector was not initialized or has been corrupted.

7 Accuracy

Not applicable.

8 Further Comments

To generate an observation, y, from the beta distribution of the second kind from an observation, x, generated by G05SBF the transformation, $y = x/(1-x)$, may be used.

9 Example

This example prints a set of five pseudorandom numbers from a beta distribution with parameters $a = 2.0$ and $b = 2.0$, generated by a single call to G05SBF, after initialization by G05KFF.

9.1 Program Text

```
*     G05SBF Example Program Text
*     Mark 22 Release. NAG Copyright 2006.
*     .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          MSTATE, MSEED, N
      PARAMETER        (MSTATE=633,MSEED=1,N=5)
*     .. Local Scalars ..
      DOUBLE PRECISION A, B
      INTEGER          GENID, I, IFAIL, LSEED, LSTATE, SUBID
*     .. Local Arrays ..
      DOUBLE PRECISION X(N)
      INTEGER          SEED(MSEED), STATE(MSTATE)
*     .. External Subroutines ..
      EXTERNAL         G05KFF, G05SBF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G05SBF Example Program Results'
      WRITE (NOUT,*)
*     Set the distribution parameters A and B
      A = 2.0D0
      B = 2.0D0
*     Initialize the seed
      SEED(1) = 1762543
*     GENID and SUBID identify the base generator
      GENID = 1
      SUBID = 1
*     Initialize the generator to a repeatable sequence
      LSTATE = MSTATE
      LSEED = MSEED
      IFAIL = 1
      CALL G05KFF(GENID,SUBID,SEED,LSEED,STATE,LSTATE,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99998) IFAIL
         GO TO 20
      END IF
*     Generate the variates
      IFAIL = 1
      CALL G05SBF(N,A,B,STATE,X,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99997) IFAIL
         GO TO 20
      END IF
*     Display the variates
      WRITE (NOUT,99999) (X(I),I=1,N)
*
   20 CONTINUE
*
99999 FORMAT (1X,F10.4)
99998 FORMAT (1X,' ** G05KFF returned with IFAIL = ',I5)
99997 FORMAT (1X,' ** G05SBF returned with IFAIL = ',I5)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 G05SBF Example Program Results

     0.5977
     0.6818
     0.1797
     0.4174
     0.4987
```